SECURITY INCIDENT REPORT

corp-dc01 / MySQL destructive compromise and unauthorized remote access

Classification: Destructive database compromise (ransomware-style extortion) with concurrent unauthorized RDP access

Severity: Critical

Evidence window: 2026-07-30 18:44:10 UTC - 2026-08-11 16:37:46 UTC

Affected asset: corp-dc01 (10.3.0.45), Windows host and MySQL service

Prepared by: Ken T.

Date: 08-14-2026

Time basis: endpoint CSV TimeGenerated values were exported in America/Los_Angeles (PDT) and converted to UTC (+7 hours). MySQL RawData timestamps were already UTC. Defender Advanced Hunting uses Timestamp; replace it with TimeGenerated when running the same hunt in Log Analytics.

1. Executive Summary

On 2026-07-31 at 01:36:02 UTC, external IP 64.89.163.79 successfully authenticated to the MySQL service on corp-dc01 as root. Within 26 seconds, the session enumerated and read kt_corp tables, issued destructive DROP statements, and created a ransom note; within two minutes it issued DROP DATABASE commands for kt_corp, sakila, and world.

Destructive root sessions recurred through 2026-08-11 from 13 source IPs in 64.89.163.0/24. Additional successful external MySQL root authentications were observed from six other sources; their relationship to the destructive campaign is Not determined from available logs. Separately, four external IPs recorded successful RemoteInteractive administrator logons; the 149.88.100.43 session was followed by Chrome and Octo Browser installation, a Defender path exclusion, and outbound Octo traffic. Data exfiltration, detection time, and completed containment are Not determined from available logs.

Common KQL parameters - prepend to each hunt

```
let START_TIME = datetime(2026-07-30T18:44:00Z);
let END_TIME = datetime(2026-08-11T16:38:00Z);
let DEVICE = "corp-dc01";
let MYSQL_RESOURCE = "/virtualmachines/corp-dc01";
let MySQLEvents = () {
    MySQLAudit_CL
    | extend ResourceId = tolower(tostring(column_ifexists("_ResourceId", ""))),
             EventText = tostring(column_ifexists("RawData", ""))
    | where ResourceId endswith MYSQL_RESOURCE
    | extend EventTimeUTC = todatetime(extract(
                 @"(\d{4}-\d{2}-\d{2}T[0-9:\.]+Z)", 1, EventText)),
             ConnectionId = extract(@"Z\s+(\d+)\s+", 1, EventText),
             Username = coalesce(
                 extract(@"Connect\s+([^@\s]+)@", 1, EventText),
                 extract(@"user\s+'([^']*)'", 1, EventText)),
             Source = coalesce(
                 extract(@"Connect\s+[^@\s]+@(\S+)\s+on", 1, EventText),
                 extract(@"user\s+'[^']*'@'([^']+)'", 1, EventText)),
             QueryText = extract(@"\sQuery\s+(.+)$", 1, EventText),
             EventClass = case(
                 EventText contains "Access denied", "LogonFailure",
                 EventText matches regex @"\sConnect\s", "LogonSuccess",
                 EventText matches regex @"\sQuery\s", "Query", "Other")
};
```

MySQL hunts use MySQLAudit_CL with TimeGenerated, _ResourceId, and RawData. Prepend the common MySQLEvents function above to each MySQL hunt. It scopes the VM through _ResourceId and parses RawData without assuming DeviceName, Username, IpAddress, ActionType, or Query exists in the live custom table.

Hunt 1A - destructive MySQL activity

```
MySQLEvents()
| where TimeGenerated between (START_TIME .. END_TIME)
| where EventText has_any ("DROP TABLE", "DROP DATABASE", "RECOVER_YOUR_DATA",
                           "bitcoin", "onionmail.org")
| project EventTimeUTC, ConnectionId, EventText
| order by EventTimeUTC asc
```

Hunt 1B - successful external endpoint logons

```
DeviceLogonEvents
| where Timestamp between (START_TIME .. END_TIME)
| where DeviceName =~ DEVICE and ActionType == "LogonSuccess"
| where LogonType in ("RemoteInteractive", "Network")
| summarize FirstSeen=min(Timestamp), Events=count()
    by RemoteIP, AccountName, LogonType
| order by FirstSeen asc
```

tTime	SessionIP	SessionUser	EventText		
Jul 30, 2026 6:35:30 PM	64.89.163.79	root	2026-07-31T01:35:30.961838Z	60 Query	USE `recover_your_data`
Jul 30, 2026 6:35:32 PM	64.89.163.79	root	2026-07-31T01:35:32.115597Z	61 Query	SELECT LOGFILE_GROUP_NAME, FILE_NAME, TOTAL_EX
Jul 30, 2026 6:35:32 PM	64.89.163.79	root	2026-07-31T01:35:32.219379Z	61 Query	SELECT DISTINCT TABLESPACE_NAME, FILE_NAME, LOC
Jul 30, 2026 6:35:32 PM	64.89.163.79	root	2026-07-31T01:35:32.410982Z	61 Init DB	recover_your_data
Jul 30, 2026 6:35:32 PM	64.89.163.79	root	2026-07-31T01:35:32.634058Z	61 Query	SELECT table_name FROM INFORMATION_SCHEMA.TA
Jul 30, 2026 6:35:32 PM	64.89.163.79	root	2026-07-31T01:35:32.730121Z	61 Query	SELECT engine, table_type FROM INFORMATION_SCHE
Jul 30, 2026 6:35:32 PM	64.89.163.79	root	2026-07-31T01:35:32.826210Z	61 Query	SELECT engine, table_type FROM INFORMATION_SCHE
Jul 30, 2026 6:35:33 PM	64.89.163.79	root	2026-07-31T01:35:33.017609Z	61 Query	select column_name, extra, generation_expression, dat
Jul 30, 2026 6:35:33 PM	64.89.163.79	root	2026-07-31T01:35:33.116187Z	61 Query	SELECT /*!40001 SQL_NO_CACHE */ `text` FROM `recov
Jul 30, 2026 6:35:33 PM	64.89.163.79	root	2026-07-31T01:35:33.753472Z	62 Connect	root@64.89.163.79 on recover_your_data using SSL/TL
Jul 30, 2026 6:35:33 PM	64.89.163.79	root	2026-07-31T01:35:33.859023Z	62 Query	SELECT * FROM recover_your_data LIMIT 10
Jul 30, 2026 6:35:34 PM	64.89.163.79	root	2026-07-31T01:35:34.010762Z	60 Query	USE `recover_your_data`
Jul 30, 2026 6:35:34 PM	64.89.163.79	root	2026-07-31T01:35:34.213112Z	60 Query	SELECT TABLE_NAME, COLUMN_NAME, CONSTRAINT_
Jul 30, 2026 6:35:34 PM	64.89.163.79	root	2026-07-31T01:35:34.314674Z	60 Query	DROP TABLE `recover_your_data`
Jul 30, 2026 6:35:34 PM	64.89.163.79	root	2026-07-31T01:35:34.417025Z	60 Query	CREATE TABLE IF NOT EXISTS RECOVER_YOUR_DATA (t
Jul 30, 2026 6:35:34 PM	64.89.163.79	root	2026-07-31T01:35:34.518314Z	60 Query	INSERT INTO RECOVER_YOUR_DATA (text) VALUES ('All
Jul 30, 2026 6:35:35 PM	64.89.163.79	root	2026-07-31T01:35:35.402115Z	64 Query	DROP DATABASE `recover_your_data`
Jul 30, 2026 6:35:35 PM	64.89.163.79	root	2026-07-31T01:35:35.898227Z	65 Query	CREATE DATABASE IF NOT EXISTS RECOVER_YOUR_DA
Jul 30, 2026 6:36:12 PM	64.89.163.79	root	2026-07-31T01:36:12.390671Z	66 Query	SHOW DATABASES
Jul 30, 2026 6:36:18 PM	64.89.163.79	root	2026-07-31T01:36:18.443293Z	70 Query	SELECT * FROM credentials LIMIT 10
Jul 30, 2026 6:36:21 PM	64.89.163.79	root	2026-07-31T01:36:21.200225Z	72 Query	SELECT * FROM customers LIMIT 10
Jul 30, 2026 6:36:23 PM	64.89.163.79	root	2026-07-31T01:36:23.929508Z	74 Query	SELECT * FROM orders LIMIT 10
Jul 30, 2026 6:36:26 PM	64.89.163.79	root	2026-07-31T01:36:26.607051Z	76 Query	SELECT * FROM payments LIMIT 10
Jul 30, 2026 6:36:27 PM	64.89.163.79	root	2026-07-31T01:36:27.649483Z	68 Query	DROP TABLE `payments`
Jul 30, 2026 6:36:27 PM	64.89.163.79	root	2026-07-31T01:36:27.766018Z	68 Query	DROP TABLE `orders`
Jul 30, 2026 6:36:27 PM	64.89.163.79	root	2026-07-31T01:36:27.880839Z	68 Query	DROP TABLE `customers`
Jul 30, 2026 6:36:27 PM	64.89.163.79	root	2026-07-31T01:36:27.999582Z	68 Query	DROP TABLE `credentials`

Figure 1. Correlated MySQL audit evidence showing activity around the first confirmed external root access from 64.89.163.79, including database enumeration, access to sensitive tables, ransom-related activity, and destructive SQL operations.

2. Incident Details

Field	Value
Detection date/time	Not determined from available logs
Reporter	Not determined from available logs
First confirmed DB access	2026-07-31 01:36:02.766791 UTC - root from 64.89.163.79
First confirmed ransom note	2026-07-31 01:36:28.243014 UTC
Classification	Unauthorized access; destructive database extortion; unauthorized remote access
Severity	Critical - privileged access plus destructive database activity

Field	Value
Affected host	corp-dc01 / 10.3.0.45
Affected data stores	kt_corp, sakila, world; attacker-created recover_your_data artifacts
Endpoint account	administrator
Database account	root

Hunt 2A - establish first external root success

```
MySQLEvents()
| where TimeGenerated between (START_TIME .. END_TIME)
| where EventClass == "LogonSuccess" and Username =~ "root"
| where isnotempty(Source) and Source !in ("localhost", "127.0.0.1", "::1")
| top 1 by EventTimeUTC asc
| project EventTimeUTC, Username, Source, Result=EventClass,
          ConnectionId, EventText
```

Hunt 2B - scope endpoint telemetry

```
union isfuzzy=true DeviceLogonEvents, DeviceProcessEvents,
    DeviceRegistryEvents, DeviceNetworkEvents, DeviceFileEvents
| where Timestamp between (START_TIME .. END_TIME)
| where DeviceName =~ DEVICE
| summarize FirstSeen=min(Timestamp), LastSeen=max(Timestamp), Events=count()
    by DeviceName
```



Figure 2. MySQL audit evidence showing the first confirmed successful external root connection to corp-dc01 from 64.89.163.79 at 2026-07-31 01:36:02 UTC, displayed as July 30, 2026 6:36:02 PM PDT in Log Analytics.

3. Impact Assessment

Area	Assessment
Confidentiality	Confirmed query access to credentials, customers, orders, and payments via SELECT * statements. Transfer of returned rows off-host is Not determined from available logs.
Integrity	The export contains 27 DROP TABLE statements and repeated DROP DATABASE statements. kt_corp, sakila, and world were targeted; kt_corp was later recreated and targeted again.
Availability	Logical database availability was disrupted by table/database deletion. Exact downtime and service-level impact are Not determined from available logs.
Scope	One Windows/MySQL host is confirmed. Lateral movement to other assets is Not determined from available logs.
Business impact	Recovery effort and loss of database availability are supported. Customer, financial, or regulatory impact is Not determined from available logs.

Hunt 3A - read and destructive SQL

```
MySQLEvents()
| where TimeGenerated between (START_TIME .. END_TIME)
| where QueryText has_any ("SELECT * FROM credentials", "SELECT * FROM customers",
                           "SELECT * FROM orders", "SELECT * FROM payments",
                           "DROP TABLE", "DROP DATABASE")
| project EventTimeUTC, ConnectionId, Sql=QueryText
| order by EventTimeUTC asc
```

Hunt 3B - network-volume check for possible exfiltration

```
NTANetAnalytics
| where TimeGenerated between (START_TIME .. END_TIME)
| where SrcIp == "10.3.0.45"
| extend OutBytes = tolong(column_ifexists("BytesSent",
                            column_ifexists("SentBytes", 0)))
| summarize Flows=count(), BytesOut=sum(OutBytes) by DestIp, DestPort
| top 25 by BytesOut desc
```

NTANetAnalytics was not included in the supplied attachments; exfiltration cannot be resolved from the present evidence package.

EventTime	ConnectionId	EventType	Sql
Jul 30, 2026 6:36:18 PM	70	Query	SELECT * FROM credentials LIMIT 10
Jul 30, 2026 6:36:21 PM	72	Query	SELECT * FROM customers LIMIT 10
Jul 30, 2026 6:36:23 PM	74	Query	SELECT * FROM orders LIMIT 10
Jul 30, 2026 6:36:26 PM	76	Query	SELECT * FROM payments LIMIT 10
Jul 30, 2026 6:36:27 PM	68	Query	DROP TABLE `payments`
Jul 30, 2026 6:36:27 PM	68	Query	DROP TABLE `orders`
Jul 30, 2026 6:36:27 PM	68	Query	DROP TABLE `customers`
Jul 30, 2026 6:36:27 PM	68	Query	DROP TABLE `credentials`
Jul 30, 2026 6:37:40 PM	79	Query	DROP TABLE `store`
Jul 30, 2026 6:37:40 PM	79	Query	DROP TABLE `staff_list`
Jul 30, 2026 6:37:58 PM	127	Query	DROP TABLE `countrylanguage`
Jul 30, 2026 6:37:58 PM	127	Query	DROP TABLE `country`
Jul 30, 2026 6:37:59 PM	127	Query	DROP TABLE `city`
Jul 30, 2026 6:37:59 PM	134	Query	DROP DATABASE `kt_corp`
Jul 30, 2026 6:38:00 PM	134	Query	DROP DATABASE `sakila`
Jul 30, 2026 6:38:00 PM	134	Query	DROP DATABASE `world`

Figure 3. MySQL audit evidence showing access to credentials, customer, order, and payment data followed by DROP TABLE and DROP DATABASE operations.

4. Indicators of Compromise

Type	Indicator	Assessment	Evidence
BTC	bc1qk9kvwhzt60u3eqcjllqlj44h0tj7w7n72apz99	Provided; not observed	Analyst-provided IOC
Email	ak+28t2@onionmail.org	Provided; not observed	Analyst-provided IOC
URL	hxxps://2no[.]co/2mysql	Provided; not observed	Analyst-provided IOC
DATAID	28T2	Provided; not observed	Analyst-provided IOC
BTC	bc1qa83x6l2dlgkx7cc9rmrymscp5sa3ljepu42w2r	Observed - high	Ransom INSERT, 2026-07-31
BTC	bc1q83wm83f3473wheapq9k2rt75kjke7nj2sy4tf6	Observed - high	Ransom INSERT, 2026-08-01 to 08-03
BTC	bc1q7jps5432akuflg9flw2vu6hgmmj5hrrdu6c5gm	Observed - high	Ransom INSERT, 2026-08-04 to 08-11
Email	ak+2jfco@onionmail.org	Observed - high	Ransom INSERT
URL	hxxps://bit[.]ly/22mysql	Observed - high	Ransom INSERT
DATAID	2JFCO	Observed - high	Ransom INSERT

Type	Indicator	Assessment	Evidence
Destructive MySQL source IPs (13)	64.89.163.77, .78, .79, .80, .89, .93, .142, .159, .164, .167, .168, .169, .178	Observed - high	root auth plus destructive bursts
Other external MySQL root sources	34.77.33.184; 34.79.30.30; 64.89.163.154; 104.236.9.211; 152.32.146.202; ec2-35-92-106-157.us-west-2.compute.amazonaws.com	Observed; campaign relationship Not determined from available logs	Successful external root authentication; enumeration confirmed for 64.89.163.154 and the EC2 hostname
RDP source IPs	80.66.83.80; 79.124.58.202; 149.88.100.43; 180.120.118.31	Observed - high	RemoteInteractive success
SHA-256	83efc72d3a3e1272037188fd7a208039 f4ec7c7add073d4a2b3a112e169e58dd	Contextual	Octo installer on host
Domains	app[.]octobrowser[.]net; app01[.]octobrowser[.]net; app[.]obiwankenode[.]com	Contextual	Octo outbound traffic

Contextual indicators identify unauthorized activity in this incident; they are not asserted to be globally malicious infrastructure or software.

Hunt 4A - endpoint IOC sweep

```
let ENDPOINT_IPS = dynamic(["80.66.83.80", "79.124.58.202",
    "149.88.100.43", "180.120.118.31"]);
union
(DeviceLogonEvents
 | where Timestamp between (START_TIME .. END_TIME)
 | where RemoteIP in (ENDPOINT_IPS)
 | project Timestamp, DeviceName, Indicator=RemoteIP, Evidence=ActionType),
(DeviceFileEvents
 | where Timestamp between (START_TIME .. END_TIME)
 | where SHA256 == "83efc72d3a3e1272037188fd7a208039f4ec7c7add073d4a2b3a112e169e58dd"
 | project Timestamp, DeviceName, Indicator=SHA256, Evidence=FileName)
| order by Timestamp asc
```

Hunt 4B - database IOC sweep

```
let DB_SOURCES = dynamic([
    "64.89.163.77", "64.89.163.78", "64.89.163.79", "64.89.163.80",
    "64.89.163.89", "64.89.163.93", "64.89.163.142", "64.89.163.154",
    "64.89.163.159", "64.89.163.164", "64.89.163.167", "64.89.163.168",
    "64.89.163.169", "64.89.163.178", "34.77.33.184", "34.79.30.30",
    "104.236.9.211", "152.32.146.202",
    "ec2-35-92-106-157.us-west-2.compute.amazonaws.com"]);
MySQLEvents()
| where TimeGenerated between (START_TIME .. END_TIME)
| where Source in (DB_SOURCES)
    or EventText has_any ("bc1q", "onionmail.org", "DATAID", "22mysql")
| project EventTimeUTC, Source, Evidence=EventText
| order by EventTimeUTC asc
```

estamp	DeviceName	IndicatorType	Indicator	Account	Evidence
Jul 30, 2026 11:43:...	corp-dc01	RDP source IP	80.66.83.80	administrator	LogonSuccess | Network
Jul 30, 2026 11:45:...	corp-dc01	RDP source IP	80.66.83.80	administrator	LogonSuccess | Network
Jul 31, 2026 11:13:...	corp-dc01	RDP source IP	80.66.83.80	administrator	LogonSuccess | Network
Jul 31, 2026 11:13:...	corp-dc01	RDP source IP	80.66.83.80	administrator	LogonSuccess | Network
Jul 31, 2026 11:13:...	corp-dc01	RDP source IP	80.66.83.80	administrator	LogonAttempted | Unknown
Jul 31, 2026 11:13:...	corp-dc01	RDP source IP	80.66.83.80	administrator	LogonSuccess | RemoteInteractive
Jul 31, 2026 11:13:...	corp-dc01	RDP source IP	80.66.83.80	administrator	LogonSuccess | RemoteInteractive
Jul 31, 2026 11:13:...	corp-dc01	RDP source IP	80.66.83.80	administrator	LogonAttempted | Unknown
Jul 31, 2026 11:13:...	corp-dc01	RDP source IP	80.66.83.80	administrator	LogonSuccess | RemoteInteractive
Jul 31, 2026 11:13:...	corp-dc01	RDP source IP	80.66.83.80	administrator	LogonSuccess | RemoteInteractive
Jul 31, 2026 12:10:...	corp-dc01	RDP source IP	79.124.58.202	administrator	LogonSuccess | Network
Jul 31, 2026 12:10:...	corp-dc01	RDP source IP	79.124.58.202	administrator	LogonAttempted | Unknown
Jul 31, 2026 12:10:...	corp-dc01	RDP source IP	79.124.58.202	administrator	LogonSuccess | RemoteInteractive
Jul 31, 2026 12:10:...	corp-dc01	RDP source IP	79.124.58.202	administrator	LogonSuccess | RemoteInteractive
Aug 1, 2026 4:04:1...	corp-dc01	RDP source IP	149.88.100.43	administrator	LogonSuccess | Network
Aug 1, 2026 4:04:2...	corp-dc01	RDP source IP	149.88.100.43	administrator	LogonSuccess | Network
Aug 1, 2026 4:04:2...	corp-dc01	RDP source IP	149.88.100.43	administrator	LogonAttempted | Unknown
Aug 1, 2026 4:04:2...	corp-dc01	RDP source IP	149.88.100.43	administrator	LogonSuccess | RemoteInteractive
Aug 1, 2026 4:04:2...	corp-dc01	RDP source IP	149.88.100.43	administrator	LogonSuccess | RemoteInteractive
Aug 1, 2026 4:06:4...	corp-dc01	SHA-256	83efc72d3a3e12720371...	administrator	FileRenamed | C:\Users\administrator\Downloads\Octo_Browser_latest_win.exe\Octo_Browser_la
Aug 1, 2026 4:11:0...	corp-dc01	SHA-256	83efc72d3a3e12720371...	administrator	FileCreated | C:\ProgramData\Caphyon\Advanced Installer\{EE009A99-B7A6-404B-83C8-DB5632
Aug 7, 2026 5:24:1...	corp-dc01	RDP source IP	180.120.118.31	administrator	LogonSuccess | Network
Aug 7, 2026 5:24:1...	corp-dc01	RDP source IP	180.120.118.31	administrator	LogonAttempted | Unknown
Aug 7, 2026 5:24:1...	corp-dc01	RDP source IP	180.120.118.31	administrator	LogonSuccess | RemoteInteractive
Aug 7, 2026 5:24:1...	corp-dc01	RDP source IP	180.120.118.31	administrator	LogonSuccess | RemoteInteractive

Figure 4. Microsoft Defender IOC matches showing administrator logons from confirmed external IP addresses and the SHA-256 hash associated with the Octo Browser installer.

5. Timeline

UTC	Phase	Event	Source
2026-07-30 18:44:10	Evidence start	First endpoint record in the supplied evidence window.	DeviceRegistryEvents
2026-07-30 19:00:25	Control change	Domain, Standard, and Public Windows Firewall profiles changed from enabled (1) to disabled (0). Initiator: svchost.exe.	DeviceRegistryEvents
2026-07-30 23:05:27	Possible endpoint access	External Network logon success for administrator from 189.186.19.250; relationship to later MySQL/RDP activity is Not determined from available logs.	DeviceLogonEvents
2026-07-31 01:36:02	Initial DB access	root successfully connected from 64.89.163.79 over TCP/IP.	MySQL auth
2026-07-31 01:36:12-26	Discovery / access	SHOW DATABASES, schema enumeration, and SELECT * against credentials, customers, orders, and payments.	MySQL query
2026-07-31 01:36:27-28	Destruction	DROP TABLE issued for payments, orders, customers, and credentials.	MySQL query
2026-07-31 01:36:28	Ransom note	RECOVER_YOUR_DATA table created; BTC, email, URL, and DATAID inserted.	MySQL query
2026-07-31 01:37:59-01:38:00	DB compromise	DROP DATABASE issued for kt_corp, sakila, and world.	MySQL query + auth

UTC	Phase	Event	Source
2026-07-31 18:13:35	RDP access	First confirmed RemoteInteractive administrator success from 80.66.83.80.	DeviceLogonEvents
2026-08-01 23:04:26	RDP access	RemoteInteractive administrator success from 149.88.100.43.	DeviceLogonEvents
2026-08-01 23:06:42-23:17:36	Post-access	Octo installer created, Octo Browser installed, Defender exclusion recorded, then outbound Octo connections began.	File / Process / Registry / Network
2026-08-01 to 2026-08-11	Recurrence	Repeated root-authenticated ransom/destructive bursts from rotating 64.89.163.x sources.	MySQL auth + query
2026-08-08 05:07:58	Rebuild activity	kt_corp recreation script created the database and tables. Authorization and completeness are Not determined from available logs.	MySQL query
2026-08-08 05:08:43	Privileged config	CREATE USER 'root'@'%' IDENTIFIED BY <secret> statement issued; query result not logged.	MySQL query
2026-08-08 07:21:02-29	Re-compromise	64.89.163.159 connected as root; kt_corp and recover_your_data were dropped.	MySQL auth + query
2026-08-11 01:24:13-25	Latest destruction	64.89.163.167 authenticated; kt_corp and recover_your_data were dropped and a ransom database recreated.	MySQL auth + query
Not determined	Detection	Detection time and detecting control are Not determined from available logs.	Evidence gap
2026-08-11 16:37:46	Evidence end	Last endpoint event in the supplied export; this is not proof of containment.	DeviceNetworkEvents

Hunt 5A - first MySQL attack sequence

```
let ATTACK_START = datetime(2026-07-31T01:36:02Z);
let ATTACK_END = datetime(2026-07-31T01:38:01Z);
MySQLEvents()
| where TimeGenerated between (ATTACK_START .. ATTACK_END)
| where EventTimeUTC between (ATTACK_START .. ATTACK_END)
| project EventTimeUTC, ConnectionId, EventText
| order by EventTimeUTC asc
```

Hunt 5B - RDP-to-Octo endpoint sequence

```
let T1 = datetime(2026-08-01T23:03:00Z);
let T2 = datetime(2026-08-01T23:20:00Z);
union
(DeviceLogonEvents
 | project Timestamp, DeviceName, Source="Logon",
           Detail=strcat(ActionType, " | ", AccountName, " | ", RemoteIP)),
(DeviceProcessEvents
 | project Timestamp, DeviceName, Source="Process",
           Detail=strcat(FileName, " | ", ProcessCommandLine)),
(DeviceFileEvents
 | project Timestamp, DeviceName, Source="File",
           Detail=strcat(ActionType, " | ", FolderPath)),
(DeviceRegistryEvents
 | project Timestamp, DeviceName, Source="Registry",
           Detail=strcat(RegistryKey, " | ", RegistryValueName)),
(DeviceNetworkEvents
 | project Timestamp, DeviceName, Source="Network",
           Detail=strcat(ActionType, " | ", RemoteIP, " | ", RemoteUrl))
| where Timestamp between (T1 .. T2) and DeviceName =~ DEVICE
| order by Timestamp asc
```

tTime	DeviceName	Source	Detail
Aug 1, 2026 4:04:18 PM	corp-dc01	Logon	administrator | Network | 149.88.100.43
Aug 1, 2026 4:04:23 PM	corp-dc01	Logon	administrator | Network | 149.88.100.43
Aug 1, 2026 4:04:26 PM	corp-dc01	Logon	administrator | RemoteInteractive | 149.88.100.43
Aug 1, 2026 4:04:26 PM	corp-dc01	Logon	administrator | RemoteInteractive | 149.88.100.43
Aug 1, 2026 4:06:42 PM	corp-dc01	File	FileRenamed | C:\Users\administrator\Downloads\Octo_Browser_latest_win.exe\Octo_Browser_latest_win.exe
Aug 1, 2026 4:07:01 PM	corp-dc01	File	FileDeleted | C:\Users\administrator\Downloads\Octo_Browser_latest_win.exe\Octo_Browser_latest_win.exe
Aug 1, 2026 4:07:02 PM	corp-dc01	Process	Octo_Browser_latest_win.exe | "Octo_Browser_latest_win.exe"
Aug 1, 2026 4:07:06 PM	corp-dc01	Process	Octo_Browser_latest_win.exe | "Octo_Browser_latest_win.exe"
Aug 1, 2026 4:07:07 PM	corp-dc01	Registry	HKEY_CURRENT_USER\S-1-5-21-3631289187-137467312-992425193-1000_Classes\Local Settings\Software\Microsoft\Windows\Shell\MuiCache |
Aug 1, 2026 4:07:07 PM	corp-dc01	File	FileCreated | C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Browser 3.0.7\install\decoder.dll\decoder.dll
Aug 1, 2026 4:07:08 PM	corp-dc01	File	FileCreated | C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Browser 3.0.7\install\2CB216C\octo.msi\octo.msi
Aug 1, 2026 4:07:14 PM	corp-dc01	Process	Octo_Browser_latest_win.exe | "Octo_Browser_latest_win.exe" /i "C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Br
Aug 1, 2026 4:07:20 PM	corp-dc01	File	FileCreated | C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Browser 3.0.7\install\2CB216C_internal\base_library.z
Aug 1, 2026 4:07:20 PM	corp-dc01	File	FileCreated | C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Browser 3.0.7\install\2CB216C_internal\webview\lib\
Aug 1, 2026 4:07:20 PM	corp-dc01	File	FileCreated | C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Browser 3.0.7\install\2CB216C_internal\assets_new\a
Aug 1, 2026 4:07:20 PM	corp-dc01	File	FileCreated | C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Browser 3.0.7\install\2CB216C_internal\assets_new\a
Aug 1, 2026 4:07:20 PM	corp-dc01	File	FileCreated | C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Browser 3.0.7\install\2CB216C_internal\assets_new\a
Aug 1, 2026 4:07:20 PM	corp-dc01	File	FileCreated | C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Browser 3.0.7\install\2CB216C_internal\assets_new\a
Aug 1, 2026 4:07:20 PM	corp-dc01	File	FileCreated | C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Browser 3.0.7\install\2CB216C_internal\assets_new\a
Aug 1, 2026 4:07:20 PM	corp-dc01	File	FileCreated | C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Browser 3.0.7\install\2CB216C_internal\assets_new\a
Aug 1, 2026 4:07:20 PM	corp-dc01	File	FileCreated | C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Browser 3.0.7\install\2CB216C_internal\assets_new\a
Aug 1, 2026 4:07:20 PM	corp-dc01	File	FileCreated | C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Browser 3.0.7\install\2CB216C_internal\assets_new\a
Aug 1, 2026 4:07:20 PM	corp-dc01	File	FileCreated | C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Browser 3.0.7\install\2CB216C_internal\assets_new\a
Aug 1, 2026 4:07:20 PM	corp-dc01	File	FileCreated | C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Browser 3.0.7\install\2CB216C_internal\assets_new\a
Aug 1, 2026 4:07:20 PM	corp-dc01	File	FileCreated | C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Browser 3.0.7\install\2CB216C_internal\assets_new\a
Aug 1, 2026 4:07:20 PM	corp-dc01	File	FileCreated | C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Browser 3.0.7\install\2CB216C_internal\assets_new\a
Aug 1, 2026 4:07:20 PM	corp-dc01	File	FileCreated | C:\Users\administrator\AppData\Roaming\Blackbird Secure Browsing LTD\Octo Browser 3.0.7\install\2CB216C_internal\assets_new\a

Figure 5. Microsoft Defender timeline showing successful administrator access from 149.88.100.43 followed by the download, execution, and installation of Octo Browser.

6. Root Cause / Attack Vector

- Confirmed database vector: internet-origin traffic reached local TCP/3306; external systems then authenticated successfully as MySQL root and executed destructive SQL.
- Confirmed endpoint vector: internet-origin traffic reached TCP/3389 and four external IPs achieved RemoteInteractive administrator logons.
- Contributing conditions: all three Windows Firewall profiles were disabled during the evidence window, and a CREATE USER 'root'@'%' statement was issued on 2026-08-08.
- How the attacker obtained or guessed the Windows administrator and MySQL root credentials is Not determined from available logs.
- No vulnerability exploit, phishing event, or confirmed malware-based initial access is present in the supplied exports. Registry persistence attributable to the attacker is Not determined from available logs.

Hunt 6A - accepted inbound MySQL and RDP connections

```
DeviceNetworkEvents
| where Timestamp between (START_TIME .. END_TIME)
| where DeviceName =~ DEVICE
| where ActionType == "InboundConnectionAccepted"
| where LocalPort in (3306, 3389)
| summarize FirstSeen=min(Timestamp), LastSeen=max(Timestamp), Connections=count()
    by LocalPort, RemoteIP, InitiatingProcessFileName
| order by FirstSeen asc
```

Hunt 6B - privileged MySQL account changes

```
MySQLEvents()
| where TimeGenerated between (START_TIME .. END_TIME)
| where EventText has_any ("CREATE USER", "ALTER USER", "GRANT")
| project EventTimeUTC, ConnectionId, EventText
| order by EventTimeUTC asc
```

Hunt 6C - Windows Firewall state changes

```
DeviceRegistryEvents
| where Timestamp between (START_TIME .. END_TIME)
| where DeviceName =~ DEVICE and RegistryValueName == "EnableFirewall"
| project Timestamp, RegistryKey, PreviousRegistryValueData,
          RegistryValueData, InitiatingProcessFileName
```



ntTime	EvidenceType	Subject	Source	Details
Jul 30, 2026 12:00:25 PM	Firewall disabled	Domain profile	svchost.exe	Enabled value changed: 1 -> 0 | HKEY_LOCAL_MACHINE\SYSTEM\ControlSet001\Services\SharedAccess\Para
Jul 30, 2026 12:00:25 PM	Firewall disabled	Standard profile	svchost.exe	Enabled value changed: 1 -> 0 | HKEY_LOCAL_MACHINE\SYSTEM\ControlSet001\Services\SharedAccess\Para
Jul 30, 2026 12:00:25 PM	Firewall disabled	Public profile	svchost.exe	Enabled value changed: 1 -> 0 | HKEY_LOCAL_MACHINE\SYSTEM\ControlSet001\Services\SharedAccess\Para
Jul 30, 2026 6:36:02 PM	Accepted inbound connection	TCP/3306	64.89.163.79	Last seen: 2026-07-31 01:36:02 UTC | Count: 1 | Process: mysqld.exe
Jul 31, 2026 11:07:54 AM	Accepted inbound connection	TCP/3389	80.66.83.80	Last seen: 2026-08-11 04:56:14 UTC | Count: 8 | Process: svchost.exe
Jul 31, 2026 12:10:22 PM	Accepted inbound connection	TCP/3389	79.124.58.202	Last seen: 2026-07-31 19:10:22 UTC | Count: 1 | Process: svchost.exe
Aug 1, 2026 4:03:19 PM	Accepted inbound connection	TCP/3389	149.88.100.43	Last seen: 2026-08-01 23:03:19 UTC | Count: 1 | Process: svchost.exe
Aug 7, 2026 5:20:52 PM	Accepted inbound connection	TCP/3389	180.120.118.31	Last seen: 2026-08-08 00:20:52 UTC | Count: 1 | Process: svchost.exe

Figure 6. Microsoft Defender evidence showing all Windows Firewall profiles disabled and accepted inbound MySQL and RDP connections from incident-associated external IP addresses.

7. Response Actions

Observed in the supplied logs

Time / item	Action	Assessment
2026-08-08 05:07:58 UTC	kt_corp recreation activity	Database and tables were created. Whether this was an approved recovery and whether data was complete are Not determined from available logs.

Time / item	Action	Assessment
2026-08-08 05:08:43 UTC	CREATE USER root@'%' statement	If successful, this permits broad remote root access. Query success is not logged; external root access recurred approximately 2 hours 12 minutes later.
Endpoint isolation / credential rotation / reimage	Not determined	No confirming event or case-management record is included in the supplied package.

Recommended response

1. Contain: isolate corp-dc01; remove public exposure of TCP/3306 and TCP/3389; preserve the VM, MySQL general/audit logs, and Defender investigation package before destructive remediation.
2. Credential control: rotate Windows administrator, MySQL root, service, and any credentials stored in the credentials table. Remove root@'%' and use a least-privileged admin account restricted by source and TLS.
3. Eradicate: reimage the Windows VM because multiple external administrator sessions occurred. If reimaging is impossible, remove Octo Browser, its Defender exclusion, and any unauthorized software only after evidence capture, then perform a full offline scan and persistence review.
4. Recover: restore kt_corp from a known-good backup predating 2026-07-31 01:36 UTC; validate schema, row counts, constraints, users/grants, and application functionality before reconnecting clients.
5. Monitor: confirm no post-containment inbound 3306/3389 access, external root success, destructive SQL, or connections from listed IOCs.

Hunt 7A - recovery and risky account changes

```
MySQLEvents()
| where TimeGenerated between (START_TIME .. END_TIME)
| where EventText has_any ("CREATE DATABASE kt_corp", "CREATE TABLE customers",
                           "CREATE TABLE credentials", "CREATE USER 'root'@'%'")
| project EventTimeUTC, ConnectionId, EventText
| order by EventTimeUTC asc
```

Hunt 7B - post-window recurrence check

```
DeviceNetworkEvents
| where Timestamp > END_TIME and DeviceName =~ DEVICE
| where ActionType == "InboundConnectionAccepted" and LocalPort in (3306, 3389)
| project Timestamp, LocalPort, RemoteIP, InitiatingProcessFileName
| order by Timestamp asc
```

EventTime	EventType	Username	SourceIP	ConnectionId	Evidence
Aug 7, 2026 10:07:...	Query			9	CREATE DATABASE kt_corp CHARACTER SET utf8mb4
Aug 7, 2026 10:07:...	Query			9	CREATE TABLE customers (
Aug 7, 2026 10:07:...	Query			9	CREATE TABLE credentials (
Aug 7, 2026 10:08:...	Query			51	CREATE USER 'root'@'%' IDENTIFIED BY <secret>
Aug 7, 2026 11:45:...	Query			96	DROP DATABASE `recover_your_data`
Aug 7, 2026 11:45:...	Query			96	DROP DATABASE `recover_your_data`
Aug 8, 2026 12:18:...	LogonSuccess	root	64.89.163.159	56	2026-08-08T07:18:20.478147Z 56 Connect root@64.89.163.159 on using TCP/
Aug 8, 2026 12:18:...	LogonSuccess	root	64.89.163.159	57	2026-08-08T07 ... using TCP/
Aug 8, 2026 12:18:...	LogonSuccess	root	64.89.163.159	58	2026-08-08T0 2026-08-08T07:18:28.327696Z 59 Connect root@64.89.163.159 on using TCP/IP using TCP/
Aug 8, 2026 12:18:...	LogonSuccess	root	64.89.163.159	59	2026-08-08T07:18:28.327696Z 59 Connect root@64.89.163.159 on using TCP/
Aug 8, 2026 12:18:...	LogonSuccess	root	64.89.163.159	60	2026-08-08T07:18:28.703975Z 60 Connect root@64.89.163.159 on using TCP/
Aug 8, 2026 12:18:...	LogonSuccess	root	64.89.163.159	61	2026-08-08T07:18:29.176887Z 61 Connect root@64.89.163.159 on using TCP/
Aug 8, 2026 12:18:...	LogonSuccess	root	64.89.163.159	62	2026-08-08T07:18:29.559631Z 62 Connect root@64.89.163.159 on using TCP/
Aug 8, 2026 12:18:...	LogonSuccess	root	64.89.163.159	63	2026-08-08T07:18:29.935775Z 63 Connect root@64.89.163.159 on using TCP/
Aug 8, 2026 12:18:...	LogonSuccess	root	64.89.163.159	64	2026-08-08T07:18:30.314592Z 64 Connect root@64.89.163.159 on using TCP/
Aug 8, 2026 12:18:...	LogonSuccess	root	64.89.163.159	65	2026-08-08T07:18:31.261650Z 65 Connect root@64.89.163.159 on using SSL/
Aug 8, 2026 12:18:...	LogonSuccess	root	64.89.163.159	66	2026-08-08T07:18:33.252118Z 66 Connect root@64.89.163.159 on recover_you
Aug 8, 2026 12:18:...	LogonSuccess	root	64.89.163.159	67	2026-08-08T07:18:34.271931Z 67 Connect root@64.89.163.159 on using TCP/
Aug 8, 2026 12:18:...	LogonSuccess	root	64.89.163.159	68	2026-08-08T07:18:34.961569Z 68 Connect root@64.89.163.159 on using TCP/
Aug 8, 2026 12:18:...	Query			68	DROP DATABASE `recover_your_data`

Figure 7. MySQL audit evidence showing kt_corp database recreation and a broad root-account statement followed by external root access from 64.89.163.159 and renewed destructive database activity.

8. Evidence

Artifact	Rows	Use	File citation identifier
DeviceLogonEvents.csv	737	Windows logon success/failure, source IP, account, logon type	file_0000000096b881fb8832c4b5bfbec337
DeviceProcessEvents(2).csv	431	Process execution and command lines	file_0000000000688230b7b496010f010c40
DeviceRegistryEvents(2).csv	14,367	Firewall state, software registration, Defender exclusion	file_00000000545c81fd9c6a1293c1e38d9b
DeviceNetworkEvents(2).csv	27,524	Inbound 3306/3389 and Octo outbound connections	file_00000000a6a481fb8a8bebc3a9f2684b
DeviceFileEvents(2).csv	15,468	Installer creation, hashes, MySQL data-file activity	file_00000000da4c81fba31a722cceb51217
mySQLAudit_CL_Auth.csv	274	MySQL success/failure, username, source IP, raw UTC record	file_0000000042b081f78961ee56aca7c25f
mySQL_Auth_CL_Query.csv	1,954	Enumeration, SELECT, DROP, ransom note, rebuild SQL	file_000000005e6881fbb09336b013bd5644
NTANetAnalytics	Not supplied	Needed for byte-volume/exfiltration assessment	Evidence gap

Key artifacts referenced: MySQL RawData UTC timestamps and connection IDs; Octo installer SHA-256; inbound connection records for local ports 3306/3389; Windows firewall registry values; and Defender path-exclusion registry activity.

Hunt 8A - endpoint evidence inventory

```
union withsource=SourceTable isfuzzy=true DeviceLogonEvents,
    DeviceProcessEvents, DeviceRegistryEvents, DeviceNetworkEvents, DeviceFileEvents
| where Timestamp between (START_TIME .. END_TIME)
| where DeviceName =~ DEVICE
| summarize Rows=count(), FirstSeen=min(Timestamp), LastSeen=max(Timestamp)
    by SourceTable
| order by SourceTable asc
```

Hunt 8B - MySQL evidence inventory

```
MySQLEvents()
| where TimeGenerated between (START_TIME .. END_TIME)
| summarize Rows=count(), FirstSeenUTC=min(EventTimeUTC), LastSeenUTC=max(EventTimeUTC)
    by EventClass
| order by EventClass asc
```



SourceTable	FirstRecord	LastRecord
DeviceFileEvents	Jul 30, 2026 11:45:06 AM	Aug 11, 2026 9:23:21 AM
DeviceLogonEvents	Jul 30, 2026 12:50:41 PM	Aug 11, 2026 9:18:42 AM
DeviceNetworkEvents	Jul 30, 2026 11:44:41 AM	Aug 11, 2026 9:37:46 AM
DeviceProcessEvents	Jul 30, 2026 11:45:13 AM	Aug 11, 2026 9:32:27 AM
DeviceRegistryEvents	Jul 30, 2026 11:44:10 AM	Aug 11, 2026 9:10:37 AM

Figure 8. Microsoft Defender evidence inventory showing the endpoint telemetry sources and collection coverage available for corp-dc01 during the incident investigation window.

9. Lessons Learned / Recommendations

Priority	Recommendation	Concrete action
P1	Eliminate public management/database exposure	Deny Internet access to 3306/3389 at the Azure NSG and host firewall. Use VPN/Bastion/JIT access and explicit allowlists.
P1	Remove broad privileged database access	Delete root@'%'; require least privilege, source restrictions, TLS, and strong unique credentials stored outside the database.

Priority	Recommendation	Concrete action
P1	Rebuild and rotate	Reimage the affected VM and rotate all Windows/MySQL/application secrets before service restoration.
P2	Alert on destructive and privileged activity	Create detections for external root success, external RemoteInteractive success, DROP DATABASE/TABLE, RECOVER_YOUR_DATA, ransom strings, and Defender exclusions.
P2	Harden recovery	Maintain immutable/offline backups and test full restore validation, including users/grants, schema, row counts, constraints, and application checks.
P3	Close evidence gaps	Retain network-flow byte counts, accurate UTC timestamps, detection/case metadata, isolation evidence, and ownership/authorization records for recovery actions.

Hunt 9A - alert candidate: failed-to-successful remote logon

```
DeviceLogonEvents
| where Timestamp > ago(1h)
| where LogonType in ("RemoteInteractive", "Network")
| summarize Failed=countif(ActionType == "LogonFailed"),
            Succeeded=countif(ActionType == "LogonSuccess"),
            FirstSeen=min(Timestamp), LastSeen=max(Timestamp)
    by DeviceName, RemoteIP, AccountName
| where Succeeded > 0 and (Failed > 0 or ipv4_is_private(RemoteIP) == false)
```

Hunt 9B - historical validation: destructive/ransom SQL

```
let VALIDATION_START = datetime(2026-07-31T01:35:30Z);
let VALIDATION_END = datetime(2026-07-31T01:38:30Z);
MySQLEvents()
| where TimeGenerated between (VALIDATION_START .. VALIDATION_END)
| where EventTimeUTC between (VALIDATION_START .. VALIDATION_END)
| extend Detection = case(
    EventText contains "DROP TABLE", "Table destruction",
    EventText contains "DROP DATABASE", "Database destruction",
    EventText has_any ("bitcoin", "onionmail.org", "DATAID"), "Ransom-note activity",
    EventText contains "RECOVER_YOUR_DATA", "Ransom artifact",
    "Other")
| where Detection != "Other"
| summarize Matches=count(), arg_min(EventTimeUTC, ConnectionId, EventText) by Detection
| project Detection, Matches, FirstDetectedUTC=EventTimeUTC,
          ConnectionId, Example=EventText
| order by FirstDetectedUTC asc
```

Matches	FirstDetectedUTC	ConnectionId	Example
9	Jul 30, 2026 6:36:27 PM	68	DROP TABLE `payments`
10	Jul 30, 2026 6:36:28 PM	68	CREATE TABLE IF NOT EXISTS RECOVER_YOUR_DATA (text VARCHAR(255))
8	Jul 30, 2026 6:36:28 PM	68	INSERT INTO RECOVER_YOUR_DATA (text) VALUES ('All your data was backed up by us. Y
3	Jul 30, 2026 6:37:59 PM	134	DROP DATABASE `kt_corp`

Figure 9. Historical validation of the proposed MySQL detection logic, showing matches for table destruction, database destruction, ransom artifacts, and ransom-note activity.

10. Geographic Attack-Source Visualization

Purpose: visualize the approximate geographic distribution of public IPv4 addresses observed in Windows/MDE and MySQL authentication telemetry during the incident window. The map is supporting scope evidence; it does not identify the physical location or identity of an attacker.

Use the following query in the Microsoft Sentinel Workbook Logs data source. It combines public Windows RemoteInteractive/Network authentication events with MySQL connection and access-denied records, aggregates them by source IP and telemetry source, and returns the latitude/longitude fields required by the Workbook Map visualization.

Hunt 10 - public authentication-source geolocation

```
let START_TIME = datetime(2026-07-30T18:44:00Z);
let END_TIME = datetime(2026-08-11T16:38:00Z);
let WindowsOrigins =
    DeviceLogonEvents
    | where TimeGenerated between (START_TIME .. END_TIME)
    | where DeviceName startswith "corp-dc01"
    | where LogonType in~ ("Network", "RemoteInteractive")
    | where ActionType in~ ("LogonSuccess", "LogonFailed")
    | where isnotempty(RemoteIP)
    | project EventTime=TimeGenerated, SourceIP=tostring(RemoteIP),
              DataSource="Windows/MDE",
              Outcome=iff(ActionType =~ "LogonSuccess", "Successful", "Failed");
let MySQLOrigins =
    MySQLAudit_CL
    | where TimeGenerated between (START_TIME .. END_TIME)
    | where _ResourceId has "corp-dc01"
    | extend EventText=tostring(column_ifexists("RawData", ""))
    | where EventText contains "Connect" or EventText contains "Access denied"
    | extend SourceIP=extract(@"@'?((?:\d{1,3}\.){3}\d{1,3})", 1, EventText),
             ParsedTime=todatetime(extract(@"(\d{4}-\d{2}-\d{2}T[0-9:\.]+Z)", 1, EventText))
    | project EventTime=coalesce(ParsedTime, TimeGenerated), SourceIP,
              DataSource="MySQL",
              Outcome=iff(EventText contains "Access denied", "Failed", "Successful");
union WindowsOrigins, MySQLOrigins
| where SourceIP matches regex @"^(?:\d{1,3}\.){3}\d{1,3}$"
| where ipv4_is_private(SourceIP) == false
| summarize AttackCount=count(),
            Successful=countif(Outcome == "Successful"),
            Failed=countif(Outcome == "Failed"),
            FirstSeen=min(EventTime), LastSeen=max(EventTime)
    by SourceIP, DataSource
| extend Geo=geo_info_from_ip_address(SourceIP)
| extend Country=tostring(Geo.country), Region=tostring(Geo.state),
         City=tostring(Geo.city), Latitude=toreal(Geo.latitude),
         Longitude=toreal(Geo.longitude)
| where isnotnull(Latitude) and isnotnull(Longitude)
| project Latitude, Longitude, SourceIP, DataSource, Country, Region, City,
          AttackCount, Successful, Failed, FirstSeen, LastSeen
| order by AttackCount desc
```

Figure 10 - Sentinel Workbook geomap



Figure 10. Microsoft Sentinel Workbook geomap showing the approximate geographic distribution of public IPv4 addresses observed in Windows/MDE and MySQL authentication telemetry for corp-dc01 during 2026-07-30 18:44 UTC through 2026-08-11 16:38 UTC. Bubble size represents authentication-event count; red markers indicate at least one successful authentication.

Interpretation limitation: IP geolocation is approximate and may identify a hosting provider, VPN, proxy, scanner, or relay rather than an operator's physical location. Failed-only sources are attack-surface telemetry, not confirmed compromise sources. Hostname-only and non-geolocatable records are omitted from this IPv4 map.

End of report. Findings are limited to the supplied exports; absence of an event is not proof that the activity did not occur.